Integration Formulas

Run Section

Instead of explicit integration, we directly replace monomials in the weak collision operator with the results of their integration over the corresponding domains.

```
(* using Gam0 prevents MMA to do strange manipulations with the Γ-function. *)
Gam0[nn_?NumberQ] := Gamma[nn];

(* ∫_0^2π ∫_0^π vx^nn vy^mm vz^pp sin(θ) dθ dφ *)
IntDirections[nn_, mm_, pp_] :=
  If[OddQ[nn] || OddQ[mm] || OddQ[pp], 0, 2 (Gam0[(nn+1)/2] × Gam0[(mm+1)/2] × Gam0[(pp+1)/2]) / Gam0[(nn+mm+pp+3)/2]];

(* ∫_0^∞ a^2nn Exp[-a^2/4] da *)
IntGauss4[nn_] := 4^nn Gam0[1/2 + nn];

(* ∫_0^∞ a^2nn Exp[-a^2/2] da *)
IntGauss2[nn_] := 2^nn/Sqrt[2] Gam0[1/2 + nn];

(* ∫_0^∞ a^2nn Exp[-a^2] da *)
IntGauss1[nn_] := 1/2 Gam0[1/2 + nn];

(* ∫_0^∞ a^nn Exp[-a] da *)
IntLaguerre[nn_] := Gam0[1 + nn];

(* ∫_0^1 r^nr (1-r)^nr1 dr *)
IntR[nr_, nr1_] := (Gam0[1 + nr] × Gam0[1 + nr1]) / Gam0[2 + nr + nr1];

(* ∫_0^∞ ∫_0^∞ ∫_0^∞ g^n (g^2/4 + ι + ι1)^m ι^α0 ι1^α1 Exp[-(g^2/4 + ι + ι1)] dg dι dι1 *)
IntEtot[n_, m_, α0_, α1_] :=
  2^n Gam0[(n + 1)/2] × Gam0[α0 + 1] × Gam0[α1 + 1] Gam0[(n+1)/2 + (α0 + 1) + (α1 + 1) + m] / Gam0[(n+1)/2 + (α0 + 1) + (α1 + 1)];
```

```
In[∘]:= DeltaValues[list_] := Block[{a = list[[1]], b = list[[2]], c = list[[3]]},
          If[OddQ[a] || OddQ[b] || OddQ[c], 0, Multinomial[a/2, b/2, c/2] / Multinomial[a, b, c]]
        ];
(* scattering tensor *)
Iσ[list_] := 4 π Bpoly DeltaValues[list] / (Total[list] + 1);
```

Collision Integral Coefficient

Run Section

This function takes three polynomials as input and computes
the entry in the Q-matrix of the collision operator. The slots ‘delta’ and ‘zeta’ receive the values of δ and ζ.

```
QcomputePoly[ψ_, ϕ1_, ϕ2_, delta_, zeta_] := Block[{δ = delta, ξ = zeta},
  (* This is the notation which we will use in the code.
     dimensionless velocities: C'/θ^(1/2)=ξ'={ξξx,ξξy,ξξz},
     C/θ^(1/2)=ξ={ξx,ξy,ξz}, C*/θ^(1/2)=ξ*={ξ1x,ξ1y,ξ1z}, C'*/θ^(1/2)=ξ'*
     and dimensionless internal energies
     I'/(m θ)=ιι, I/(m θ)=ι, I*/(m θ)=ι1, I'*/(m θ)=ιι1
     and the center of mass peculiar velocity  h:=(ξ+ξ*)/2,
     relative velocity g:=|u|=|ξ-ξ*| and gp:=|u'|=|ξ'-ξ'*|. *)

  (* raw original polynomial expression in variables (ξx,ξy,ξz,ξ1x,ξ1y,ξ1z,ξξx,ξξy,ξξz) *)
  expression1 = 1/2 (ψ[ξξx, ξξy, ξξz, ιι] - ψ[ξx, ξy, ξz, ι]) ×
    (ϕ1[ξx, ξy, ξz, ι] × ϕ2[ξ1x, ξ1y, ξ1z, ι1] + ϕ2[ξx, ξy, ξz, ι] × ϕ1[ξ1x, ξ1y, ξ1z, ι1]);

  (*  From mass conservation law  ξ'* + ξ'=ξ+ξ* and
      energy conservation law  E=1/2|ξ|^2+1/2|ξ*|^2+ι+ι1 = 1/2|ξ'|^2+1/2|ξ'*|^2+ιι+ιι1,
     together with parameters r,R,σ, one can express pre-collisional velocities as
      function of post-collisional quantities.
      Also polar coordinates for relative velocity u are introduced as u=g{νx,νy,νz}*.
      We use the abbreviation R1:= 1-R, r1:= 1-r   *)

  (* two substitutions map the polynomials into variables (hx,hy,hz,g,νx,νy,νz,σx,σy,σz) *)
  subAfterCollision =
   {ξξx → hx + 1/2 gp σx, ξξy → hy + 1/2 gp σy, ξξz → hz + 1/2 gp σz, ιι → r R1 (1/4 g^2 + ι + ι1)};

  subCenterDiff = {ξx → hx + 1/2 g νx, ξy → hy + 1/2 g νy,
    ξz → hz + 1/2 g νz, ξ1x → hx - 1/2 g νx, ξ1y → hy - 1/2 g νy, ξ1z → hz - 1/2 g νz};

  (* substitutions for the collision model *)
  subFactor = {Bpoly :→ Bhat (2 Gamθ[δ + 3/2])/(√π Gamθ[δ/2]^2) R1 R^(1/2) (r R1)^(δ/2-1) (r1 R1)^(δ/2-1)};

  subCrossSection = {Bhat :→ B ((R g^2)^(ξ/2) + η ((r R1 ι)^(ξ/2) + (r1 R1 ι1)^(ξ/2)))};

  (* substitutions for the frozen case *)
  subFrozen1 = {η → ηf, B → B/((2 Gamθ[δ+3/2])/(√π Gamθ[δ/2]^2) R1 R^(1/2) (r R1)^(δ/2-1) (r1 R1)^(δ/2-1))};

  subFrozen2 = {r → ι/(ι + ι1), r1 → ι1/(ι + ι1), R → Etot^-1 g^2/4, R1 → Etot^-1 (ι + ι1)};

  subgp = {gp → 2 Sqrt[R (1/4 g^2 + ι + ι1)]};
```

```
(* the substitutions really blow up the expression *)
expression2 = Expand[expression1 /. subAfterCollision /. subCenterDiff];

(* step-by-step we can replace monomial expressions. We start with (σx,σy,σz). *)
rule = {nn_?NumericQ, mm_, pp_} :→ Iσ[{nn, mm, pp}];

expression3 =
 Expand[Total[CoefficientRules[expression2, {σx, σy, σz}] /. rule /. Rule → Times]];

(* next monomials are (hx,hy,hz). *)
rule = {nn_?NumericQ, mm_, pp_} :→ IntGauss1[(nn + mm + pp)/2 + 1] × IntDirections[nn, mm, pp];

expression4 =
 Expand[Total[CoefficientRules[expression3, {hx, hy, hz}] /. rule /. Rule → Times]];

(* finally we replace monomials in (νx,νy,νz)
 and introduce the transition probability factor and gp *)
rule = {nn_?NumericQ, mm_, pp_} :→ g^2 IntDirections[nn, mm, pp];

expression5 =
 Expand[Total[CoefficientRules[expression4, {νx, νy, νz}] /. rule /. Rule → Times] /.
    subFactor /. subgp];

(* here we subsitute the collision model *)
expression6 = PowerExpand[expression5 /. subCrossSection];

(* we proceed with computing both the frozen and non-frozen case.
   The substitutions must be tweaked such that all terms get covered. *)
expression7frozen = Expand[PowerExpand[expression6 /. subFrozen1 /. subFrozen2]];

expression8frozen = Expand[Expand[Etot^2 ι ι1 expression7frozen] /.
   {Etot^mm_. g^nn_. ι^α0_. ι1^α1_. :→ IntEtot[nn, mm - 2, α0 - 1, α1 - 1]}];

expression7 = Expand[Expand[r r1 R R1 expression6] /. {r^nn_. r1^mm_. :→ IntR[nn - 1, mm - 1]} /.
   {R^nn_. R1^mm_. :→ IntR[nn - 1, mm - 1]}];

expression8 = Expand[
  Expand[Etot^2 ι ι1 expression7] /. {Etot^mm_. g^nn_. ι^α0_. ι1^α1_. :→ IntEtot[nn, mm - 2, α0 - 1, α1 - 1]}];

(* the result is averaged with ω *)
Simplify[ExpandAll[ω expression8 + (1 - ω) expression8frozen]]
]
```

Compute Exemplary Production Coefficients

Run Section

We follow the example from the paper.

```
In[∘]:= (* this is our choice of parameters *)
delt = 2;
zet = 1/2;
```

```
In[∘]:= (* these are the polynomials that generate the moments density, dynamic pressure,
           xx-stress component and translational/internal x-heat flux. *)
```

$$\psi = \left\{1, \frac{\delta\,(\xi x^2 + \xi y^2 + \xi z^2) - 6\,\iota}{3\,(3+\delta)}, \xi x^2 - \frac{1}{3}\,(\xi x^2 + \xi y^2 + \xi z^2), \frac{1}{2}\,\xi x\,(\xi x^2 + \xi y^2 + \xi z^2), \xi x\,\iota\right\};$$

```
(* these are the corresponding polynomials in the distribution function times the prefactor
   of the equilibrium distribution. *)
```

$$\phi = \left\{1, \frac{1}{2}\left(\xi x^2 + \xi y^2 + \xi z^2 - \frac{6\,\iota}{\delta}\right), \frac{1}{2}\,(\xi x^2 - \xi z^2), \xi x\left(\frac{1}{5}\,(\xi x^2 + \xi y^2 + \xi z^2) - 1\right), \frac{\xi x\,(2\,\iota - \delta)}{\delta}\right\}\frac{\iota^{\delta/2-1}}{2\,\sqrt{2}\,\pi^{3/2}\,\mathrm{Gam0}[\delta/2]};$$

```
n = Length[ψ];
```

```
In[∘]:= (* We compute the complete set of coefficients. Takes approximately 10 sec.
           Actually, this could be accelerated by using the known structure *)
result = Table[
                 Simplify[
                      QcomputePoly[
                      {ξx, ξy, ξz, ι} ↦ Evaluate[ψ[[i]] /. {δ → delt}],
                      {ξx, ξy, ξz, ι} ↦ Evaluate[ϕ[[j]] /. {δ → delt}],
                      {ξx, ξy, ξz, ι} ↦ Evaluate[ϕ[[k]] /. {δ → delt}], delt, zet]]
               , {i, 1, n}, {j, 1, n}, {k, 1, n} ];
```

```
In[∘]:=
```

The following is because:

- for a trace-free σ we have

$\sum_{i,j} \sigma_{\mathrm{ij}}\,\sigma_{\mathrm{ij}} = 2\,\sigma_{\mathrm{xx}}\,\sigma_{\mathrm{xx}} + \ldots$

- and

$\sum_k \sigma_{k<i}\,\sigma_{j>k} = \frac{1}{3}\,\sigma_{\mathrm{xx}}\,\sigma_{\mathrm{xx}} + \ldots$

- and for any two vector u and v we have

$u_{<i}\,v_{j>} = \frac{2}{3}\,u_x\,v_x + \ldots$

```
In[∘]:= result[[2, 3, 3]] = result[[2, 3, 3]] / 2;
result[[3, 3, 3]] = result[[3, 3, 3]] / (1/3);
result[[3, 4, 4]] = result[[3, 4, 4]] / (2/3);
result[[3, 5, 5]] = result[[3, 5, 5]] / (2/3);
result[[3, 4, 5]] = result[[3, 4, 5]] / (2/3);
result[[3, 5, 4]] = result[[3, 5, 4]] / (2/3);
```

```
In[∘]:=
```

```
In[∘]:= (* the value for stress in case of the frozen, reduced collisions serves as scale *)
fac = -2 Expand[FullSimplify[result[[3, 1, 3]]] /. {ω → 0, ηf → 0}]
```

Out[∘]= $\frac{4}{75}\,B\,\sqrt{2\,\pi}\;\mathrm{Gamma}\left[\frac{23}{4}\right]$

In[•]:=

In[•]:=
```
(* these are the coeficients printed in the paper *)
rawCoefs = Expand[FullSimplify[
    -{2 result[[2, 1, 2]], result[[2, 2, 2]],
     result[[2, 3, 3]], result[[2, 4, 4]], result[[2, 5, 5]], 2 result[[2, 4, 5]],
     2 result[[3, 1, 3]], 2 result[[3, 2, 3]], result[[3, 3, 3]], result[[3, 4, 4]],
     result[[3, 5, 5]], 2 result[[3, 4, 5]],
     2 result[[4, 1, 4]], 2 result[[4, 1, 5]], 2 result[[4, 2, 4]], 2 result[[4, 2, 5]],
     2 result[[4, 3, 4]], 2 result[[4, 3, 5]],
     2 result[[5, 1, 5]], 2 result[[5, 1, 4]], 2 result[[5, 2, 5]],
     2 result[[5, 2, 4]], 2 result[[5, 3, 5]], 2 result[[5, 3, 4]]
    } / fac /. {ηf → h / (1 - ω)} ] ];
```

In[•]:=

In[•]:=
```
(* name of productions *)
names = {P_Π^(0), P_Π^(1), P_Π^(2), P_Π^(3), P_Π^(4), P_Π^(5), P_σ^(0), P_σ^(1), P_σ^(2), P_σ^(3), P_σ^(4), P_σ^(5),
    P_q^(0), P_q^(1), P_q^(2), P_q^(3), P_q^(4), P_q^(5), P_s^(0), P_s^(1), P_s^(2), P_s^(3), P_s^(4), P_s^(5)};

(* this turns the real-valued factors into rationals *)
res = Rationalize[N[rawCoefs], 10^-6];
```

In[•]:=
```
(* ... and this gives the table as published *)
Transpose[{names, res /. {h → ηf (1 - ω)}}] // TableForm
```

Out[•]//TableForm=

P^0_Π	$\frac{929\,\omega}{876}+\frac{717\,\eta\,\omega}{649}$
P^1_Π	$\frac{149\,\omega}{1124}-\frac{383\,\eta\,\omega}{1981}$
P^2_Π	$\frac{7\,\omega}{990}$
P^3_Π	$\frac{3\,\omega}{943}$
P^4_Π	0
P^5_Π	$-\frac{5\,\omega}{943}$
P^0_σ	$1+\frac{395}{569}\eta f\,(1-\omega)-\frac{183\,\omega}{1463}+\frac{478\,\eta\,\omega}{649}$
P^1_σ	$\frac{11}{48}-\frac{220}{857}\eta f\,(1-\omega)-\frac{121\,\omega}{1010}-\frac{286\,\eta\,\omega}{2071}$
P^2_σ	$\frac{1}{21}-\frac{7}{484}\eta f\,(1-\omega)-\frac{13\,\omega}{794}$
P^3_σ	$\frac{3}{280}-\frac{3}{922}\eta f\,(1-\omega)-\frac{5\,\omega}{1204}$
P^4_σ	$-\frac{5}{672}+\frac{11\,\eta f\,(1-\omega)}{1638}+\frac{5\,\omega}{672}$
P^5_σ	$-\frac{1}{280}+\frac{1}{922}\eta f\,(1-\omega)+\frac{\omega}{280}$
P^0_q	$\frac{2}{3}+\frac{367}{793}\eta f\,(1-\omega)+\frac{319\,\omega}{1181}+\frac{1154\,\eta\,\omega}{1343}$
P^1_q	$-\frac{631\,\omega}{1190}-\frac{659\,\eta\,\omega}{1193}$
P^2_q	$\frac{1}{24}-\frac{97}{706}\eta f\,(1-\omega)+\frac{25\,\omega}{322}-\frac{324\,\eta\,\omega}{2011}$
P^3_q	$\frac{5}{72}-\frac{188}{821}\eta f\,(1-\omega)-\frac{119\,\omega}{1300}+\frac{29\,\eta\,\omega}{540}$
P^4_q	$\frac{1}{15}-\frac{18}{889}\eta f\,(1-\omega)-\frac{47\,\omega}{1065}$
P^5_q	$-\frac{1}{72}+\frac{79\,\eta f\,(1-\omega)}{1725}+\frac{32\,\omega}{551}-\frac{23\,\eta\,\omega}{1499}$
P^0_s	$\frac{115}{144}+\frac{279}{305}\eta f\,(1-\omega)+\frac{57\,\omega}{181}+\frac{1241\,\eta\,\omega}{1049}$
P^1_s	$-\frac{298\,\omega}{1405}-\frac{213\,\eta\,\omega}{964}$
P^2_s	$\frac{371}{2365}-\frac{39}{955}\eta f\,(1-\omega)-\frac{35\,\omega}{792}-\frac{109\,\eta\,\omega}{877}$
P^3_s	$\frac{253}{2688}-\frac{16}{653}\eta f\,(1-\omega)-\frac{191\,\omega}{1801}+\frac{29\,\eta\,\omega}{700}$
P^4_s	$\frac{63}{1004}-\frac{9}{551}\eta f\,(1-\omega)-\frac{39\,\omega}{788}+\frac{29\,\eta\,\omega}{1050}$
P^5_s	$\frac{12\,\omega}{943}$